# $Id: PKGBUILD 85602 2013-03-03 21:28:43Z arodseth $
# Maintainer: Alexander Rødseth <rodseth@gmail.com>
# Contributor: kappa <kappacurve@gmail.com>

pkgname=erlang-sdl
pkgver=1.2
pkgrel=7
arch=('x86_64' 'i686')
pkgdesc='SDL and OpenGL bindings for Erlang'
url='http://esdl.sourceforge.net/'
license=('custom')
depends=('erlang' 'sdl' 'glu' 'libgl' 'erlang')
makedepends=('mesa-libgl')
replaces=('esdl')
source=("$pkgname-$pkgver.tar.gz::https://github.com/dgud/esdl/archive/master.tar.gz")
sha256sums=('6a18010e12eed2abeb9b9893a2d89586e3ff14b44fce9c15b8ed091b67709438')

build() {
  cd "$srcdir/esdl-master"

  find -type f -print0 | xargs -0 chmod 644
  make
}

package() {
  cd "$srcdir/esdl-master"

  make INSTALLDIR="$pkgdir/usr/lib/erlang/lib/esdl-$pkgver" install
  install -Dm644 license.terms "$pkgdir/usr/share/licenses/$pkgname/LICENSE"
}

# vim:set ts=2 sw=2 et:
